// SPDX-License-Identifier: MIT
pragma solidity 0.7.6;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a >= b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow, so we distribute
        return (a / 2) + (b / 2) + ((a % 2 + b % 2) / 2);
    }
}

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies in extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return _functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        return _functionCallWithValue(target, data, value, errorMessage);
    }

    function _functionCallWithValue(address target, bytes memory data, uint256 weiValue, string memory errorMessage) private returns (bytes memory) {
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: weiValue }(data);
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}


/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return payable(msg.sender);
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}


 /* @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

interface IUniswapV2Router02 {
    function WETH() external pure returns (address);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
}

contract PolkazeckStake is Ownable {
    using SafeMath for uint256;
    
    uint256 constant DECIMALS = 10 ** 18;
    uint256 constant DIVISOR = 10 ** 10;
    uint256 constant STAKE_DURATION = 31540000;
    
    uint256 public allocation = 40000000 * DECIMALS;
    uint256 public maxStake = 500000 * DECIMALS;
    uint256 public minStake = 10000 * DECIMALS;
    uint256 public roiPerSeconds = 17361; // 0.15 / 1 day * DIVISOR;
    uint256 public totalStaked;
    uint256 public totalStakers;
    uint private unlocked = 1;

    IERC20 public stakeToken;
    IERC20[] public rewardToken;
    IUniswapV2Router02 public router;

    struct Stake {
        uint256 createdAt;
        uint256 amount;
        IERC20 rewardMode;
        uint256 lastWithdrawal;
    }
    
    mapping(address => Stake) stakes;
    
    
    modifier lock() {
        require(unlocked == 1, "PolkazeckStake: LOCKED");
        unlocked = 0;
        _;
        unlocked = 1;
    }

    constructor() {
        stakeToken = IERC20(0xeDB7b7842F7986a7f211d791e8F306C4Ce82Ba32);
        router = IUniswapV2Router02(0x7a250d5630B4cF539739dF2C5dAcb4c659F2488D);
    }
    
    receive() payable external {
        revert();
    }
    
    function created(address _staker) external view returns(uint256) {
        return stakes[_staker].createdAt;
    }
    
    function staked(address _staker) external view returns(uint256) {
        return stakes[_staker].amount;
    }
    
    function rewardMode(address _staker) external view returns(IERC20) {
        return stakes[_staker].rewardMode;
    }
    
    function lastWithdrawal(address _staker) external view returns(uint256) {
        return stakes[_staker].lastWithdrawal;
    }
    
    function newStake(uint256 _amount, IERC20 selectedRewardToken) public lock {
        require(stakes[_msgSender()].amount == 0, "newStake: Staking");
        require(totalStaked.add(_amount) <= allocation, "newStake: Filled!");
        require(_amount <= maxStake, "newStake: Above maximum");
        require(_amount >= minStake, "newStake: Below minimum");
        require(isRewardToken(selectedRewardToken), "newStake: Reward not available");
        
        uint256 initialBalance = stakeToken.balanceOf(address(this));
        
        require(stakeToken.transferFrom(_msgSender(), address(this), _amount), "newStake: Transfer failed");
        
        uint256 latestBalance = stakeToken.balanceOf(address(this));
        uint256 amount = latestBalance.sub(initialBalance);
        
        stakes[_msgSender()] = Stake({createdAt: block.timestamp, amount: amount, rewardMode: selectedRewardToken, lastWithdrawal: block.timestamp});
        
        totalStakers = totalStakers.add(1);
        totalStaked = totalStaked.add(amount);
        
        emit NewStake(_msgSender(), address(selectedRewardToken), amount);
    }
    
    function _withdraw() internal {
        Stake storage stake = stakes[_msgSender()];
        if (stake.amount > 0 && stake.createdAt.add(STAKE_DURATION) > stake.lastWithdrawal) {
            uint256 thisReward = _roi(stake);
            // thisReward to rewardMode
            uint256[] memory toReward = toRewardMode(thisReward, address(stake.rewardMode));
            uint256 currentReward = toReward[toReward.length - 1];
            require(stake.rewardMode.transfer(_msgSender(), currentReward), "Withdraw: Transfer failed");
            stake.lastWithdrawal = block.timestamp;

            emit Withdraw(_msgSender(), address(stake.rewardMode), currentReward);   
        }
    }
    
    function _exit() internal {
        Stake storage stake = stakes[_msgSender()];
        require(stake.amount > 0, "_exit: !Staking");
        require(stakeToken.transfer(_msgSender(), stake.amount), "_exit: Transfer failed");
        totalStaked = totalStaked.sub(stake.amount);
        totalStakers = totalStakers.sub(1);
        stake.amount = 0;
        emit Exit(_msgSender());
    }
    
    function withdraw() public lock {
        _withdraw();
    }
    
    function exit() public lock {
        _withdraw();
        _exit();
    }
    
    function emergencyExit() public lock {
        /*
        * Exit without rewards
        */
        _exit();
    }
    
    function roi(address _staker) public view returns(uint256) {
        Stake memory stake = stakes[_staker];
        return _roi(stake);
    }
    
    function _roi(Stake memory _stake) internal view returns(uint256) {
        uint256 periodBoundary = Math.min(block.timestamp, _stake.createdAt.add(STAKE_DURATION));
        uint256 thisRewardPeriod = periodBoundary.sub(_stake.lastWithdrawal);
        return _stake.amount.mul(thisRewardPeriod).mul(roiPerSeconds).div(DIVISOR);
    }
    
    function toRewardMode(uint256 _amount, address _token) public view returns(uint256[] memory amounts) {
        address weth = router.WETH();
        address[] memory path1 = new address [](2);
        address[] memory path2 = new address [](3);
        
        if (_token == weth) {
            path1[0] = address(stakeToken);
            path1[1] = weth;
            amounts = toRewardToken(_amount, path1);
        } else {
            path2[0] = address(stakeToken);
            path2[1] = weth;
            path2[2] = _token;
            amounts = toRewardToken(_amount, path2);
        }
    }
    
    function toRewardToken(uint256 _amount, address[] memory path) public view returns(uint256[] memory amounts) {
        amounts = router.getAmountsOut(_amount, path);
    }
    
    function estimateReward(uint256 _amount) public view returns(uint256) {
        return _amount.mul(STAKE_DURATION).mul(roiPerSeconds).div(DIVISOR);
    }
    
    function isRewardToken(IERC20 _token) public view returns(bool valid) {
        valid = false;
        for (uint i = 0; i < rewardToken.length; i++) {
            if (rewardToken[i] == _token) {
                valid = true;
                break;
            }
        }
    }
    
    function getAsset(IERC20 _tokenAddress, uint256 _amount) public onlyOwner {
        require(_tokenAddress != stakeToken, "getAsset: Not allowed!");
        require(_tokenAddress.balanceOf(address(this)) >= _amount, "getAsset: Not enough balance");
        _tokenAddress.transfer(owner(), _amount);
        
        emit Withdraw(_msgSender(), address(_tokenAddress), _amount);   
        
    }
    
    function setMaxStake(uint256 _max) external onlyOwner {
        maxStake = _max;
    }
    
    function setMinStake(uint256 _min) external onlyOwner {
        minStake = _min;
    }
    
    function setRoiPerSeconds(uint256 _roiPerSeconds) external onlyOwner {
        roiPerSeconds = _roiPerSeconds;
    }

    function setAllocation(uint256 _allocation) external onlyOwner {
        allocation = _allocation;
    }

    function addRewardToken(IERC20 _token) external onlyOwner {
        rewardToken.push(_token);
    } 
    
    event NewStake(address indexed staker, address indexed selectedRewardToken, uint256 amount);
    event Withdraw(address indexed staker, address indexed rewardToken, uint256 reward);
    event Exit(address indexed staker);
}